

HANDS-ON COMPUTING WITH ORANIG PART W: A Short-run Simulation:

A Five Percent Reduction in Real Wages

You are now going to perform a short run comparative static experiment simulating the effects of a five percent reduction in real wages for all occupations in all industries.

In WinGEM, select from the *Simulation* option in the main menu.

Simulation | GEMSIM Solve...

The Command file ORANIGSR.CMF contains instructions for GEMSIM. **Select** the CMF file, ORANIGSR.CMF (an abbreviation for ORANIG Short Run). Look at ORANIGSR.CMF before you run the simulation. Choose the button **Edit**.

The short-run closure is described in section 6 of the ORANIG document and is listed in Table 3.

The solution method is **Johansen**. This gives a simple **one-step** approximation of the model's non-linear equations. The simple one-step is easier for an initial interpretation of the results.

method = Johansen ;

A better solution method is **Gragg** with a 2 4 6 multi-step solution followed by extrapolation. This produces results that are a good solution of the model's non-linear equations. However we have commented out this solution method and steps by placing an exclamation mark ! at the start of these lines.

! method = Gragg ; ! alternative to above
! steps = 2 4 6 ;

We suggest you try both solution methods and compare the results. The Command file ORANIGSR.CMF gives the Johansen solution. To use the Gragg solution method, comment out the Johansen line with an exclamation mark and remove the exclamation marks from the Gragg lines and save with a different name ORANIGSR_G.CMF. Run the two simulations, then open both Solution files at once in ViewSOL.

Remember that the Johansen solution gives a set of results which satisfy the linearised equations in the TAB file exactly and the Gragg solution results do not satisfy the linear equations exactly. However the Gragg solution is a better solution to the non-linear equations of the model. The non-linear equations are the equations we really want to solve.

In analysing the results below, use the Johansen simulation ORANIGSR.SL4.

In the Command file ORANIGSR.CMF, the updated data file is called <cmf>.UPD. This means that the updated data file will be called ORANIGSR.UPD. This data file depends on the data and the simulation we are running.

The verbal description indicates that a short run simulation is being performed with a wage-cut shock.

In the Command file ORANIGSR.CMF that contains the Short-run closure, the list of exogenous variables is written out explicitly instead of using “swap” statements. For a short-run closure, the capital stock *xlcap* is exogenous and not shocked. This means that the capital stock is fixed for a short-run simulation. Check that the variable *realwage* is exogenous. You can only shock exogenous variables.

The shocked variable is *realwage*. From the equations contained in ORANIG.TAB, what is the meaning of a 5 percent reduction in *realwage*? Look at the equation E_realwage.

Equation E_realwage $realwage = p1lab_io - p3tot$;

How does *realwage* relate to the variable *f1lab_io*? Look at Equations E_p1lab and E_p1lab_io.

```
Equation E_p1lab      # Flexible setting of money wages #
  (all,i,IND) (all,o,OCC)
    p1lab(i,o)= p3tot + f1lab_io + f1lab_o(i) + f1lab_i(o) + f1lab(i,o);

Equation E_p1lab_io
  V1LAB_IO*p1lab_io = sum{i,IND, sum{o,OCC, V1LAB(i,o)*p1lab(i,o) }};
```

In Equation E_p1lab, the variables *f1lab_i*, *f1lab_o* and *f1lab* are exogenous and not shocked so they have a value of zero. So this equation means that all values of the variable *p1lab(i,o)* for all industries *i* and occupation *o* are the same. For all industries and all occupations,

$p1lab(i,o) = p3tot + f1lab_io$

Hence, from Equation E_p1lab_io, the share-weighted sum *p1lab_io* also equals this value:

$p1lab_io = p3tot + f1lab_io$

From equation E_realwage, *f1lab_io* must be the same as *realwage*.

Now exit from the editor and **Run** GEMSIM with the CMF file ORANIGSR.CMF.

Check the results in ViewSOL – this is a simple way to start looking at the changes and percentage changes in the solution.

In a later session you will learn to use the AnalyseGE tool – useful when you want to track down individual variables or coefficients or when you want to look at the relative size of terms in an equation.

A GUIDE TO THE INTERPRETATION OF THE REAL-WAGE-CUT SIMULATION

INTRODUCTION

Now that you have run a simulation and produced a solution, you will probably feel overwhelmed by the volume of numbers the model produces. The question you now face is how to make sense of the output.

We have prepared a series of questions that guide you to the answers of some interesting results. The questions are not exhaustive and you may think of many others that are of interest.

The questions we are asking below are designed to rationalise particular results in terms of the model's theoretical framework and underlying database. Often, it is difficult to unravel all of the mechanisms leading to a particular result. However, you need to attempt to rationalise results within the context of the main identifiable mechanisms of the model. This process, apart from giving you insights into a particular economic phenomenon, serves as an informal verification of the simulation's results.

The experiment you are analysing is based on a shock to a macroeconomic variable; the real wage. You may have noticed in the closure of the model that many macro variables are on the exogenous list. For example, on the supply-side of the economy, we fixed the level of capital usage by fixing capital in each industry (x1cap). On the demand side, we fixed domestic absorption (i.e., Gross National Expenditure: real household consumption, x3tot; aggregate real investment, x2tot_i, aggregate real other demands, x5tot; and aggregate real inventories). Therefore, in the closure for the real wage experiment, we have largely imposed the macroeconomic environment.

The constraints that our choice of macro environment places on the economy will be important in determining relative price changes, and therefore the responses of agents to the effects of the cut in real wages. It is handy if we keep the nature of the macro environment in the back of our minds when attempting to interpret the results. Figure 1 presents a schematic representation of the short-run macro environment.

In Figure 1, exogenous variables are depicted in rectangles and endogenous variables are depicted in ovals. The arrows indicate a plausible direction of causation between variables. Thus, on the supply-side of the macroeconomy, we have exogenised the capital stock, technology and the real wage. With the real wage given, the model can determine aggregate employment (why?). With employment, technology and capital determined, the model can determine aggregate output (GDP).

On the demand side, aggregate household consumption, investment, other demands (and inventories) are fixed. With GDP determined from the supply side and domestic absorption (household consumption, investment, government consumption and inventories) given, the trade balance must act as an endogenous 'swing' variable to satisfy the GDP identity. That is, if as a result of our shock GDP increases/decreases relative to domestic absorption, the trade balance must move toward surplus/deficit.




Figure 1. Schematic Representation of the Short-run closure of ORANI

We will interpret the results of the real wage cut beginning with the impact on some main macroeconomic variables. This will enable us to determine the consequences of the shock for the main endogenous macroeconomic variables, and in particular, the trade balance. This will give us insights into broad (economy-wide) relative price movements that result from the shock. A knowledge of the main relative price movements will help us interpret the industry results.

Fill in the following table with the results from this simulation:

Description	Variable	Value
Real household consumption	x3tot	
Real investment	x2tot_i	
Real government demands	x5tot	
Export volume	x4tot	
Import volume CIF	x0cif_c	
Real GDP	x0gdpexp	
Aggregate capital stock	x1cap_i	
Aggregate employment	employ_i	
Absorption price index*	p0gne	
GDP price index	p0gdpexp	
CPI	p3tot	
Exports price index	p4tot	
Real devaluation*	p0realdev	
Average nominal wage	p1lab_io	
Average real wage	realwage	
Contribution of BOT to real GDP	contBOT	
Terms of trade*	p0toft	
Aggregate primary factor use	x1prim_i	

MACROECONOMIC RESULTS

Determining the impact on aggregate employment and output

1. First, check the results to see if the shock has been implemented correctly. You should check the value of the shocked variable *realwage*
the value of the variable *f1lab_io*
the economy-wide nominal wage *p1lab_io*
What is the difference between the values of *p1lab_io* and the CPI *p3tot*.
What are the values of *p1lab*?
2. What effect do you expect the reduction of the real wage to have on aggregate employment? Verify your intuition by checking the value of aggregate employment *employ_i*.
3. There are two variables in the model for real value added, commonly known as GDP at factor cost: *x1prim_i* and *xgdppfac*. Are these two values equal? Look at the equations for these variables.

From the change in aggregate employment, can you verify the change in real aggregate value added *xgdppfac*? Hint: find the equation for *xgdppfac* that calculates the change in aggregate value added using the share weighted average of the percentage changes in real factor inputs (i.e., labour, capital and land).

$$xgdppfac = SL * employ_i + SC * x1cap_i + SN * x1lnd_i$$

where SL, SC and SN are the shares of labour, capital and land in aggregate value added. How do you calculate the share of the wage bill in aggregate value added?

[Hint: in looking at the data yesterday, you found V1LND_I(2736), V1LAB_IO(194981), V1CAP_I(180575).]

4. You will have noticed that both real GDP at factor cost *xgdppfac* and real GDP at market prices *x0gdpinc* have increased, but with real GDP at factor cost rising by more than real GDP at market prices. The difference between the two aggregates is indirect taxes. The dominant effect on aggregate output is the change in employment and this causes GDP to increase (see equations determining *contGDPinc* and *x0gdpinc*).

Determining the composition of final demand

Now you have explained the change in GDP from the income side. Next we will attempt to explain the changes in the expenditure-side components of GDP.

Given the changes in real GDP and the components of national domestic absorption, what must happen to the change in the foreign trade balance? (Hint: be careful, don't use the variable *delb* which is the ordinary change in the nominal trade balance to GDP ratio. We wish to know the change in the real trade balance. The formula for the ordinary change in the real trade balance is

$$(V4TOT * x4tot - V0CIF_C * x0cif_c) / V0GDPEXP.$$

This will give you the ordinary change in the real trade balance in basecase values).

Hint: Use Summary.har to look at the values of V4TOT, V0CIF_C and V0GDPEXP. Then calculate the expression $(V4TOT * x4tot - V0CIF_C * x0cif_c) / V0GDPEXP$.

Look at the variables *contGDPexp* and *contBOT*.

5. How is the trade balance being moved? By a change in imports (*x0cif_c*) or a change in exports (*x4tot*) or by some combination?
6. Movements in the international trade balance occur due to activity effects and relative price effects. Changes in domestic demand (with given prices) will tend to change the demand for imports - an

activity effect. Changes in international competitiveness (changes in domestic costs relative to foreign prices/costs) will change foreign export demands and domestic demand for imports. How do you think exports and imports are responding in this simulation? What is the activity effect? Hint: imports are demanded for intermediate use as well as for final use. What is the relative price effect? Hint, check the variable *p0realdev*.

Now we turn our attention to the results at the industry level.

INDUSTRY RESULTS

7. Look at the industry output (*x1tot*) and employment responses(*employ*). Note some of the outstanding results (spend about 10 minutes).

You have probably noticed that some industries have performed better than others; the impact of the real wage shock changes the industrial structure of the economy. We will now attempt to explain the impact of the real wage shock on the industrial structure exploiting insights gained from our interpretation of the macro results.

8. In our analysis of the macro economy, we noticed a move towards trade surplus. We argued that the movement to trade surplus required an improvement in international competitiveness, i.e., a reduction in domestic costs relative to foreign prices. We might speculate that the improvement in competitiveness would favour the traded goods industries, i.e., those industries that sell a large share of their output to foreigners and/or which compete in domestic markets with imports. Can you verify that the traded goods industries do well (*x1tot* and *employ*)? Using ViewHAR to look at the header array file SUMMARY.HAR, check sales shares of industries contained in the header "SALE" to further identify export oriented industries, import competing industries and non-traded-sector industries.

You should also view the Fan decomposition, which you will see by clicking open the variable *fandecomp* using ViewSQL. The first column of numbers in the Fan decomposition (LocalMarket) shows us by how much we would expect local-commodity output to change, if output of the local commodity increased in line with the change in domestic demand for the commodity regardless of source (i.e., domestic or imported). The second column of numbers (DomShare) can be interpreted as the amount by which local-commodity output changes due to a relative price change favouring import replacement. The third column of numbers (Export) shows the contribution to the change in the output of the local commodity, brought about by the change in exports. The last column (Total) is the sum of the values in columns 1 to 3. For example, the increase in *TexteisCalc* output is 7.3 per cent (total column of the Fan decomposition). Column 1 of the Fan decomposition can be interpreted as saying that given the increase in domestic demand for *TexteisCalc* (local and imported), we may have anticipated the rise in output to be 2.9 per cent. However, column 2 can be interpreted as saying that due to a relative price change favouring locally produced *TexteisCalc*, output of the domestic *TexteisCalc* industry increased by an additional 0.4 percentage points (over the growth in local demand). The third column shows that increased export demand accounted for 4.0 percentage points of the total expansion in *TexteisCalc* production.

9. Can you explain any anomalies? That is, do you find any seemingly traded-goods industries that perform poorly or non-traded goods industries that perform well? Hint: remember the underlying input-output linkages in the model.
10. Can you explain the performance of the construction (*ProdConstCiv*) industry? Hint: what is its sales pattern, i.e., who are its main customers and for what purpose do they buy the construction commodity? (Try to explain why demand is moving the way it is.)
11. Can you explain the performance of the government industry (*AdmPublica*)?

```

! Command file for ORANIG model: short-run closure

check-on-read elements = warn; ! very often needed
cpu=yes ; ! (Optional) Reports CPU times for various stages
log file = yes; ! Optional
auxiliary files = ORANIG; ! needed by GEMSIM

! Solution method
!method = GRAGG ;
!steps = 2 4 6 ;
method = johansen; ! alternative to above

! Data and summary file
file basedata = BaseData.har ;
updated file basedata =<cmf>.upd;
file summary = summary.har;

! Closure! Exogenous variables constraining real GDP from the supply side
exogenous x1cap ! all sectoral capital
            x1lnd ! all sectoral agricultural land
            alcap allab_o allnd
            alprim altot a2tot ! all technological change
            realwage ; ! Average real wage

! Exogenous settings of real GDP from the expenditure side
exogenous x3tot ! real private consumption expenditure
            x2tot_i ! real investment expenditure
            x5tot ! real government expenditure on goods
            delx6 ; ! real demands for inventories by commodity

! The demand curves of exports are fixed in both quantity and price axes
exogenous f4p f4q ! individual exports
            f4p_ntrad f4q_ntrad ; ! collective exports
! Exogenous foreign prices of imports ;
exogenous pf0cif ;

! All tax rates are exogenous
exogenous delPTXRATE f0tax_s fltax_csi f2tax_csi f3tax_cs f5tax_cs t0imp
            f4tax_trad f4tax_ntrad floct ;

! distribution of government demands !
exogenous f5 ;

! The nominal exchange rate is the numeraire
exogenous phi ;

! Number of households and their consumption preferences are exogenous
exogenous q a3_s ;

exogenous capslack; ! switch off aggregate capital constraint !

! Distribution of investment between industries
xSet EXOGINV # 'exogenous' investment industries #
(SIUP, ProdConstCiv, ServFinanceI, AlugImoveis, AdmPublica);
xSubset EXOGINV is subset of IND;
xSet ENDOGINV # 'endogenous' investment industries # = IND - EXOGINV;
exogenous finv1(ENDOGINV); ! investment linked to profits
exogenous finv2(EXOGINV); ! investment follows aggregate investment

! Exogenous variables for regional extension
exogenous freg1 freg2 freg3 freg4 freg5 freg6;
exogenous rsum1 rsum2 rsum3 rsum4 rsum5 rsum6;

rest endogenous ;

verbal description =
Wage cut: DPSV shortrun closure;

shock realwage = -5 ; ! real wage shift variable

```